

Sumario

UT 03: Elementos de un programa informático.....	2
1 Ejercicios básicos.....	2
1.1 Proceso Hola.....	2
2 Condicionales.....	2
3 Bucles.....	4
3.1 Calificaciones.....	4
4 Condiciones múltiples (Según).....	6
5 Tarea entregada en el aula virtual – Programa IRPF.....	6
6 Mejoras en el código.....	7
6.1 Programa "edades".....	7
7 Subprocesos.....	10

UT 03: Elementos de un programa informático

Relación de ejercicios hechos en clase con el editor de pseudocódigo PSEInt

1 Ejercicios básicos

En este documento recogemos los ejercicios desarrollados en clase para aprender conceptos básicos de Programación.

1.1 Proceso Hola

```
Proceso UT03_Ejemplo01_ProcesoNombre
    Definir nombre Como Cadena;
    Escribir "¿Cómo te llamas? ";
    Leer nombre;
    Escribir "Hola " Sin Saltar;
    Escribir nombre;
FinProceso
```

2 Condicionales

```
Proceso UT03_Ejemplo02_Programa_edad
    Definir edad Como Entero;
    Escribir "Introduzca su edad: ";
    Leer edad;
    Si edad >= 18 Entonces
        Escribir "Bienvenido!!!!";
    SiNo
        Escribir "No puedes pasar";
    FinSi
FinProceso
```

```
Proceso UT03_Ejemplo03_Programa_edad
// Versión 2 – Condicional anidado
Definir edad Como Entero;

Escribir "Introduzca su edad: ";

Leer edad;

Si edad==18 Entonces
    Escribir "Enhorabuena, ya eres mayor!!!!";
SiNo
    // Esto es una condición anidada
    Si edad>18 Entonces
        Escribir "Bienvenido!!!";
    SiNo
        Escribir "No puedes pasar.";
    FinSi
FinSi

FinProceso
```

```
Proceso UT03_Ejemplo04_Programa_edad
// Versión 3 – Condicionales sucesivos
Definir edad Como Entero;

Escribir "Introduzca su edad: ";

Leer edad;

Si edad==18 Entonces
    Escribir "Enhorabuena, ya eres mayor!!!!";
FinSi

Si edad>=18 Entonces
    Escribir "Bienvenido!!!";
FinSi

Si edad<18 Entonces
    Escribir "No puedes pasar.";
FinSi

FinProceso
```

3 Bucles

3.1 Calificaciones

Realizamos dos versiones de un programa para calcular las calificaciones de una clase:

```
Proceso UT03_Ejemplo05_Calificaciones1
    Definir nombre como Cadena;
    Definir nota como Entero;

    Repetir
        Escribir "Dime tu nombre: ";
        Leer nombre;

        Escribir "Dime tu nota: ";
        Leer nota;

        Escribir "Tienes un " Sin Saltar;

        si nota<5 Entonces
            escribir " Suspenso";
        FinSi

        si nota>=5 y nota<7 Entonces
            escribir " Aprobado";
        FinSi

        si nota>=7 & nota<9 Entonces
            escribir " Notable";
        FinSi

        si nota>=9 Entonces
            escribir " Sobresaliente";
        FinSi

    Hasta Que Falso
FinProceso
```

```
Proceso UT03_Ejemplo06_Calificaciones2
// Bucle con una condición de finalización
Definir nombre como Cadena;
Definir nota como Real;
Definir pregunta como Cadena;
Definir otravez Como Logico;

Repetir
    Escribir "Dime tu nombre: ";
    Leer nombre;
    Escribir "Dime tu nota (usar . para decimales: )";
    Leer nota;
    Escribir "Tienes un " Sin Saltar;

    si nota<5 Entonces
        escribir " Suspenso";
    SiNo
        si nota<7 Entonces
            escribir " Aprobado";
        SiNo
            si nota<9 Entonces
                escribir "Notable";
            SiNo
                escribir "Sobresaliente";
            FinSi
        FinSi
    FinSi

    otravez <- Verdadero;

    Escribir "¿Introducir otro alumno(si/no)? ";
    Leer pregunta;

    //Diferentes posibilidades
    // Opción 1
    //si pregunta=="no" Entonces
        //otravez <- Falso;
    //FinSi

    // Opción 2
    si pregunta=="si" Entonces
        otravez <- Verdadero;
    SiNo
        otravez <- Falso;
    FinSi

Hasta Que otravez==Falso
FinProceso
```

4 Condiciones múltiples (Según)

```
Proceso UT03_Ejemplo07_Aficiones
    definir aficion Como Entero;
    Escribir "De las siguientes opciones:";
    Escribir "1 - Deportes";
    Escribir "2 - Música";
    Escribir "3 - Tumbing";
    Escribir "Dime cuál es tu afición principal: " Sin Saltar;
    Leer aficion;

    Segun aficion Hacer
        1: Escribir "Deberías estudiar más";
        2: Escribir "Deberías salir más al aire libre";
        3: Escribir "Eres un vago";
        De Otro Modo:
            Escribir "Tienes que aclarar tu vida.";
    FinSegun
FinProceso
```

5 Tarea entregada en el aula virtual – Programa IRPF

```
Proceso UT03_Ejemplo08_IRPF
    Definir sueldo Como Real;
    Definir impuesto Como Real;

    Repetir
        Escribir "¿Cuánto dinero ganas en un año?";
        Leer sueldo;

        Si sueldo<=12000 Entonces
            impuesto<-0;
        SiNo
            Si sueldo<=20000 Entonces
                impuesto<-10;
            SiNo
                Si sueldo<=30000 Entonces
```

```
                impuesto<-20;
            sino
                Si sueldo<=40000 Entonces
                    impuesto<-30;
                sino
                    Si sueldo>40000 Entonces
                        impuesto<-45;
                    FinSi
                FinSi
            FinSi
        Finsi
    FinSi
    Escribir "Pagas ", (sueldo*impuesto/100);
    Hasta Que Falso
FinProceso
```

6 Mejoras en el código

6.1 Programa "edades"

Para calcular la edad media de una serie, en diferentes versiones. Primero con "Mientras":

Proceso UT03_Ejemplo09_Edades

```
//El programa escuchará una serie de valores (edades)
//hasta que el usuario introduzca un valor negativo o cero
//y a continuación se mostrará el promedio
Definir acumulador Como Real;
Definir contador Como Entero;
Definir edad Como Entero;
acumulador<-0;
contador<-0;

Repetir
    Escribir "Introduzca edad (0 o negativa para finalizar): " Sin
Saltar;
    Leer edad;
    Si edad>0 Entonces
        acumulador<-acumulador+edad;
        contador<-contador+1;
    FinSi
Hasta Que edad<=0;

Escribir "Edad media: ",acumulador/contador;

FinProceso
```

Esta versión incluye la protección para evitar división por cero (si no se introduce ningún valor):

Proceso UT03_Ejemplo10_Edades_Mientras

```
//El programa escuchará una serie de valores (edades)
//hasta que el usuario introduzca un valor negativo o cero
//y a continuación se mostrará el promedio
Definir acumulador Como Real;
Definir contador Como Entero;
Definir edad Como Entero;
acumulador<-0;
contador<-0;

Escribir "Introduzca edad (0 o negativa para finalizar): " Sin
Saltar;
Leer edad;

Mientras edad>0 Hacer
    acumulador<-acumulador+edad;
    contador<-contador+1;
    Escribir "Introduzca edad (0 o negativa para finalizar): " Sin
Saltar;
    Leer edad;
FinMientras

Si contador>0 Entonces
    Escribir "Edad media: ",acumulador/contador;
SiNo
    Escribir "No has metido ningún valor, colega.";
FinSi

FinProceso
```

Esta versión con un bucle "Repetir":

Proceso UT03_Ejemplo11_Edades_Repetir

```
//El programa escuchará una serie de valores (edades)
//hasta que el usuario introduzca un valor negativo o cero
//y a continuación se mostrará el promedio
```



```
Definir acumulador Como Real;
Definir contador Como Entero;
Definir edad Como Entero;
acumulador<-0;
contador<-0;

Repetir
    Escribir "Introduzca edad (0 o negativa para finalizar): " Sin
Saltar;
    Leer edad;
    Si edad>0 Entonces
        acumulador<-acumulador+edad;
        contador<-contador+1;
    FinSi
Hasta Que edad<=0;

Si contador>0 Entonces
    Escribir "Edad media: ",acumulador/contador;
SiNo
    Escribir "No has metido ningún valor, colega.";
FinSi

FinProceso
```

Esta versión con un bucle "Para":

Proceso UT03_Ejemplo12_Edades_Para

```
//El programa escuchará una serie de valores (edades)
//hasta que el usuario introduzca un valor negativo o cero
//y a continuación se mostrará el promedio
Definir acumulador Como Real;
Definir contador Como Entero;
Definir alumnos Como Entero;
Definir edad Como Entero;
acumulador<-0;

alumnos<-0;
Mientras alumnos<=0 Hacer
    Escribir "¿Cuántos alumnos hay en clase? " Sin Saltar;
    Leer alumnos;
FinMientras

Para contador<-1 hasta alumnos Con Paso 1 Hacer
    Escribir "Introduzca edad (0 o negativa para finalizar): " Sin
```

```
Saltar;  
    Leer edad;  
    acumulador<-acumulador+edad;  
FinPara  
    Escribir "Edad media: ",acumulador/alumnos;  
FinProceso
```

7 Subprocesos

A partir del programa anterior ("Edades") esta versión incluye mejoras como protecciones ante errores y la creación de un Subproceso para calcular la media aritmética de un array:

Proceso UT03_Ejemplo13_Edades_Para_Mejorado

```
//El programa escuchará una serie de valores (edades)  
//hasta que el usuario introduzca un valor negativo o cero  
//y a continuación se mostrará el promedio  
  
//Esta versión permitirá almacenar las edades en un array  
//y calculará la media con un Subproceso  
  
Definir contador Como Entero;  
Definir alumnos Como Entero;  
Definir edad Como Entero;  
  
//Array para guardar las edades. Debe tener un tamaño fijo.  
//Lo dimensionamos a 30, que sería el máximo de alumnos  
//permitido por nuestro programa.  
Definir edad_alumno como Entero;  
Dimension edad_alumno[30];  
  
alumnos<-0;  
Mientras alumnos<=0 o alumnos>30 Hacer  
    Escribir "¿Cuántos alumnos hay en clase? " Sin Saltar;  
    Leer alumnos;  
    si alumnos<=0 o alumnos>30 Entonces  
        Escribir "Número de alumnos incorrecto. Inténtalo otra  
vez.";  
    FinSi  
FinMientras  
  
//A continuación llamamos al subproceso que recoge datos
```

```
    entradaDatos(edad_alumno,alumnos);

    //A continuación llamamos al subproceso que calcula la media
    Escribir "Edad media: ", CalculoMedio(edad_alumno,alumnos);

    Escribir "Listado de edades:";
    Para contador<-0 hasta alumnos-1 Hacer
        Escribir edad_alumno[contador]," - " Sin Saltar;
    FinPara
    Escribir "";

FinProceso

//Sacamos el código de cálculo de la media a un Subproceso
SubProceso mediaEdades <- CalculoMedio (edad_alumno,alumnos)
    Definir contador Como Entero;
    Definir mediaEdades como Real;
    mediaEdades<-0;

    Para contador<-0 hasta alumnos-1 Hacer
        mediaEdades<-mediaEdades+edad_alumno[contador];
    FinPara
    mediaEdades<-mediaEdades/alumnos;

FinSubProceso

//Sacamos el código de entrada de datos a un subproceso
Subproceso entradaDatos(edad_alumno,alumnos)
    Definir contador,edad Como Entero;

    Para contador<-0 hasta alumnos-1 Con Paso 1 Hacer
        Escribir "Introduzca edad: " Sin Saltar;
        edad<-0;
        Mientras edad<=0 Hacer
            leer edad;
            si edad<=0 Entonces
                Escribir "Edad incorrecta, introdúzcala de nuevo: "
Sin Saltar;
            FinSi
        FinMientras
        edad_alumno[contador]<-edad;
    FinPara

FinSubProceso
```